Please use your own paper to answer the following questions. Point values are shown in parentheses.

1. (10) Consider the following segment of code:

```vbnet
If txtANSICode.Text < "0" Or txtANSICode.Text > "255" Then
    MsgBox "ANSI codes must be between 0 and 255"
    Exit Sub
Else
    txtANSICChar.Text = Chr$(Val(txtANSICode.Text))
End If
```

Assume that the user enters a valid number, i.e., legal digits. Why does the code fail to achieve its goal of validating the user input? How would you correct the problem?

2. (8) The following code defines three click-event procedures and declares four variables in the General Declaration section of a form module.

```vbnet
Option Explicit

Dim W As String
Dim X As String
Dim Y As String
Dim Z As String

Private Sub Command1_Click()
    X = Text1.Text
End Sub

Private Sub Command2_Click()
    Y = Text1.Text & Z
End Sub

Private Sub Command3_Click()
    Z = Text2.Text
    W = Left$(Z, 1)
End Sub
```

Move the variable declarations to their appropriate locations assuming your objective is to limit their scope to the minimum possible and still have the program work as it did before the move.
3. (8) You have four Single type variables A, B, C, and D with values: A=2, B=5, C=10, D=50. What values are computed for the following expressions?

   a. \( A + B \mod C \)
   b. \( A + B \times C + D / C \)
   c. \( B - C - B \mod C \)
   d. \( B^A + D / B / B^2 \)

4. (9) What output is produced by the message boxes in the code that follows when the button cmdExam is clicked?

```vba
' General Declaration Section
Option Explicit
Dim X As Integer
Dim Y As Integer

Public Sub DoA(ByVal Z As Integer)
    Dim X As Integer
    Z = 100
    X = 50
End Sub

Public Sub DoB(X As Integer, ByVal Z As Integer)
    X = X + Z
    Z = Z + Y
End Sub

Public Sub DoC()
    Dim A As Integer
    Dim Y As Integer
    Dim Z As Integer
    A = 11
    Y = 20
    Z = 30
    DoB Y, Z
End Sub

Private Sub cmdExam_Click()
    Dim A As Integer
    Dim B As Integer
    Dim Z As Integer
    A = 10
    B = 20
    X = 115
    Y = -30

    DoA A
    MsgBox Str$(A) & Str$(X) & Str$(Z)

    DoB B, X
    MsgBox Str$(A) & Str$(B) & Str$(X)

    DoC
```
MsgBox Str$(A) & Str$(Y) & Str$(Z)

End Sub

5. (10) Write a KeyPress event for the text box named txtName that beeps and rejects all keystrokes that are not capital letters (A-Z), lowercase letters (a-z), the space character, or the backspace character.

6. (12) Write a function that computes and returns the present value (of type Currency) of a given future amount of money. The function should be passed the future value (Currency), the number of years (Integer), the number of times the interest is computed each year (Integer), and the annual interest rate (Single).

The formula for present value is:

\[ PV = \frac{FV}{\left(1 + \frac{r}{m}\right)^{n \times m}} \]

where:  
FV is the future value,  
n is the number of years,  
m is the number of times interest is compounded per year, and  
r is the annual interest rate.

Be sure to specify that the function cannot modify the four parameter values.
Answer Key

1. The problem is a code such as 65 (a very valid ANSI code value) will be rejected by the IF statement as an invalid code falling outside the range. This is because the comparison is a string comparison. In a string comparison, "65" is greater than "255" because the characters are compared one by one beginning with the first (left-most) character. Since "6" comes after "2" in the ANSI table, "65" is greater than "255".

To fix the problem, turn the comparison into a numeric comparison with the Val function as follows:

```vbscript
If Val(txtANSICode.Text) < 0 Or Val(txtANSICode.Text) > 255 Then
    MsgBox "ANSI codes must be between 0 and 255"
    Exit Sub
Else
    txtANSICode.Text = Chr$(Val(txtANSICode.Text))
End If
```

2.

```vbscript
Option Explicit
Dim Z As String

Private Sub Command1_Click()
    Dim X As String
    X = Text1.Text
End Sub

Private Sub Command2_Click()
    Dim Y As String
    Y = Text1.Text & Z
End Sub

Private Sub Command3_Click()
    Dim W As String
    Z = Text2.Text
    W = Left$(Z, 1)
End Sub
```

3. a. 2
   b. 57
   c. -5
   d. 25.4

4. First message box: 10 115 0
   Second message box: 10 135 115
   Third message box: 10 -30 0
5.  

Private Sub txtName_KeyPress(KeyAscii As Integer)

Select Case KeyAscii
Case Asc("A") To Asc("Z"), Asc("a") To Asc("z"), Asc(" "), 8
    'OK - pass through
Case Else
    Beep
    KeyAscii = 0
End Select
End Sub

6.

Public Function PresentValue(ByVal FV As Currency, 
    ByVal n As Integer, ByVal m As Integer, 
    ByVal r As Single) As Currency

    PresentValue = FV / (1 + r / m) ^ (n * m)
End Function